Special FIBPlus features: network traffic optimization and FIBPlus Repository

Network traffic optimization

In this article we will try to give a number of recommendations and examples, which will enable developers to create more effective Firebird and InterBase applications. These technologies of optimization of network traffic can be applied to multi-user environment and systems that address to the server using low-speed channels (i.e. ISDN or dial-up).

Caching metadata

FIBPlus enables us to get system information on table columns and automatically set properties of such TpFIBDataSet fields as Required (for NOT NULL fields), ReadOnly (for calculated ones) and DefaultExpression (for those with the default option). It is very convenient for a developer and a user as the first does not have to set the properties manually during the development of the client application and the second gets more informative messages when working with the program. For example, if a database field is described as NOT NULL, and the user will try to leave it empty, he or she will get the message «Field ‘…’ must have a value». Such a message is more understandable than system Firebird and InterBase error of the PRIMARY KEY violation. The same concerns calculated fields, as you surely cannot edit them. FIBPlus will automatically set their ReadOnly properties to True and users will not see incomprehensible errors when trying to change the field properties in TDBGrid. 

But this FIBPlus feature has a disadvantage that reveals during work with low-speed connection channels. In order to get information on fields FIBPlus components execute additional "invisible" queries addressing to system Firebird and InterBase tables. Of course if there are many tables in the application or many fields in these tables, speed of application work may slow down and the net traffic may increase. In particular you can see this on the stage of first query openings as each of them entails a number of additional ones. Then during work with recurrent queries FIBPlus uses information it has previously got but at the start of the application you may notice a small slowing down of work. 

At first sight the developer has only two solutions of this problem: either to tolerate delays on first opening of queries or not to use this nice function of FIBPlus components, which will set all table fields on its own. But actually you can use such FIBPlus feature as caching metadata on the client.

Now we will demonstrate this in practice. You create a new application and place there the following components:

pFIBDatabase1: TpFIBDatabase;

pFIBTransaction1: TpFIBTransaction;

pFIBDataSet1: TpFIBDataSet;

DBGrid1: TDBGrid;

DataSource1: TDataSource;

FIBSQLMonitor1: TFIBSQLMonitor;

Label1: TLabel;

Label2: TLabel;

Label3: TLabel;

Memo1: TMemo;

Button1: TButton;

ListView1: TListView;

And then link the components in the following way:

pFIBDatabase1.DefaultTransaction := pFIBTransaction1;

pFIBTransaction1.DefaultDatabase := pFIBDatabase1;

pFIBDataSet1.Database := pFIBDatabase1;

pFIBDataSet1.Transaction := pFIBTransaction1;

pFIBDataSet1.AutoCommit := True;

DBGrid1.DataSource := DataSource1;

DataSource1.DataSet := pFIBDataSet1; 

Set parameters of connection to a database, use SQL Generator to create queries in pFIBDataSet1:

SELECT

    EMP.EMP_NO,

    EMP.FIRST_NAME,

    EMP.LAST_NAME,

    EMP.PHONE_EXT,

    EMP.HIRE_DATE,

    EMP.DEPT_NO,

    EMP.JOB_CODE,

    EMP.JOB_GRADE,

    EMP.JOB_COUNTRY,

    EMP.SALARY,

    EMP.FULL_NAME

FROM

    EMPLOYEE EMP

Now let's write handlers of OnFormCreate (for the Form1) and OnSQL events (for the FIBSQLMonitor1 component):

procedure TForm1.FIBSQLMonitor1SQL(EventText: String;

  EventTime: TDateTime);

begin

  Memo1.Lines.Text := Memo1.Lines.Text + EventText;

end;

 

procedure TForm1.FormCreate(Sender: TObject);

begin

  pFIBDatabase1.Connected := True;

  pFIBDataset1.Active := True;

end;

Now it is time to look at the ListView1 component. It is necessary to add four columns there:  

Name, Not Null, Computed and Default. They are necessary to check if properties of pFIBDataset1 columns have been set correctly. For this we will write the AfterOpen even handler of pFIBDataset1:

procedure TForm1.pFIBDataSet1AfterOpen(DataSet: TDataSet);

var FieldInfo: TListItem;

    Index: Integer;

begin

  with pFIBDataSet1 do begin

    for Index := 0 to pred(FieldCount) do begin

      FieldInfo := ListView1.Items.Add;

      FieldInfo.Caption := Fields[Index].FieldName;

      if Fields[Index].Required then FieldInfo.SubItems.Add('+')

      else FieldInfo.SubItems.Add('-');

      if Fields[Index].ReadOnly then FieldInfo.SubItems.Add('+')

      else FieldInfo.SubItems.Add('-');

      if Fields[Index].DefaultExpression <> '' then

         FieldInfo.SubItems.Add(Fields[Index].DefaultExpression);

    end;

  end;

end;

So in result our application will look as: 

Now start the application and pay attention at contents of ListView1:

As shown in the picture, pFIBDataSet1 has got all properties of table fields and even default expressions. This was achieved by use of a number of additional queries hooked by FIBSQLMonitor, for example:

[Application: metadata_cache.exe]

: [Execute] Select R.RDB$FIELD_NAME,R.RDB$FIELD_SOURCE,F.RDB$COMPUTED_BLR, 

R.RDB$DEFAULT_SOURCE DS,F.RDB$DEFAULT_SOURCE DS1, 

F.RDB$FIELD_TYPE, 

F.RDB$CHARACTER_SET_ID 

from  RDB$RELATION_FIELDS R 

JOIN RDB$FIELDS F ON (R.RDB$FIELD_SOURCE = F.RDB$FIELD_NAME) 

where  R.RDB$RELATION_NAME=:TN 

order by R.RDB$FIELD_POSITION

  TN = 'EMPLOYEE'

Execute tick count 0

It is obvious that if our application had a hundred of tables, we would get a hundred of additional queries. During work in the net these queries do not cause any problems, but if you have a low speed channel they slow down speed of work. 

Now we will speak about the main issue, which concerns caching metadata on the client. You should set the following values of the CacheSchemeOptions property of pFIBDatabase1: 

Now pFIBDatabase1 will save all obtained metadata (in our case they are properties of such fields as NOT NULL, COMPUTED, etc) in the external file «metadata_cache.fpc» on closing the application and load them from this file on opening it. So on recurrent opening of application there will be no need to use additional queries. There is also one thing concerning caching metadata: pFIBDatabase1 will try to check validity of current metadata. For this it will again make a query to system tables and check if the system table identifier has been changed and the structure updated. If any of the table structures on the server has been changed, its data will be reread. Of course this will not affect other tables. 

You can avoid such a default check if you write your own realization in the OnAcceptCacheSchema handler of pFIBDatabase1. In the simplest case we will say that all saved metadata are always valid:

procedure TForm1.pFIBDatabase1AcceptCacheSchema(const ObjName: String;

  var Accept: Boolean);

begin

  Accept := True;

end;

This handler is called for metadata of each object saved in the file. Notice that if real metadata differ from those cached by our application, there may appear difficult-to-find errors. Then start our application. After the first start you will not see any changes, as the application will execute all the queries as earlier. But if you start the program again you will see that Memo1 no more has queries to RDB$RELATION_FIELDS. Actually the application will only execute one query: 

[Application: metadata_cache.exe]

pFIBDataSet1.SelectQuery: [Execute] SELECT

    EMP.EMP_NO,

    EMP.FIRST_NAME,

    EMP.LAST_NAME,

    EMP.PHONE_EXT,

    EMP.HIRE_DATE,

    EMP.DEPT_NO,

    EMP.JOB_CODE,

    EMP.JOB_GRADE,

    EMP.JOB_COUNTRY,

    EMP.SALARY,

    EMP.FULL_NAME

FROM

    EMPLOYEE EMP

Though there are no system queries, all properties of fields in pFIBDataset1 (as shown in ListView1) will be correct. We wanted such a result in order to reduce unnecessary queries and at the same time to continue using nice FIBPlus features.

Dynamic modifying queries

Before telling about modifying queries we will explain how user's data modification by TpFIBDataSet works. The query used to get data is set in the SelectSQL property. On inserting a record (to be more exact when the Post method is called after Append/Insert) there is executed a query set in InsertSQL, and instead of parameters there are inserted real values of fields set by the user (or generated by the program). The same happens on changing a record: after calling Post, TpFIBDataSet executes a query from the UpdateSQL property inserting values there from record columns. To illustrate this we will write a small example. First we will create the following table and generator for the primary key:

CREATE TABLE "Simple Table" (

    "Id" INTEGER NOT NULL,

    "First Name" VARCHAR (100),

    "Last Name" VARCHAR (100),

    "Address" VARCHAR (100));

 

ALTER TABLE "Simple Table" ADD CONSTRAINT "PK_Simple Table" PRIMARY KEY ("Id");

 

CREATE GENERATOR “Simple Table_Id_GEN”;

Now we will create a new application. There we will place the same components as in the previous one except for TListView (we do not need it anymore). Using SQL Generator we will form a query for SelectSQL:

SELECT

    Sim."Id",

    Sim."First Name",

    Sim."Last Name",

    Sim."Address"

FROM

    "Simple Table" Sim

and then generate modifying queries. The whole process was described in the previous example. For instance, for UpdateSQL we will get the following query:

UPDATE "Simple Table" SET 

    "Id" = ?"Id",

    "First Name" = ?"First Name",

    "Last Name" = ?"Last Name",

    "Address" = ?"Address"

 WHERE     

            "Id" = ?"OLD_Id"

Now we have to set auto-generation of primary key values for the “Id” field. We will use the AutoUpdateOptions property: 

[image: image4.png]Object Inspector
rEDasser
Properies | Everts |
ol pdatepions
AucPaansTofieks
Auchewitesds

ConChange5aLs
Generatarame

GeneralarSten

KeyFields
ParamsToFieldsLinks
SeparateBloblipdate
UpdateOriyMoiiedFields
UpdateTableName
WhenGetGenD

Al shown

Fale
True

True

Simple Table_Id_GEN
1

If]
(TPatams_Fields_Links)
Fale

True

“Simple Table"
waOnNewRecord





Here there are three important properties: GeneratorName (a name of the generator), KeyFields (a name of the key field) and WhenGetGenID (an option of getting generator values). The WhenGetGenID property is set equal wgOnNewRecord, in order to let pFIBDataSet1 get a new generator value just on inserting a record. Then we start the application and add a couple of records to the table:

Now if we have a look at Memo1, we will see how the records were added:

[Application: update_only_modified]

pFIBDataSet1: [Execute] INSERT INTO "Simple Table"(

    "Id",

    "First_Name",

    "Last_Name",

    "Address"

)

VALUES(

    ?"Id",

    ?"First_Name",

    ?"Last_Name",

    ?"Address"

)

 

  Id = 1

  First_Name = 'Name 1'

  Last_Name = 'Last Name 1'

  Address = 'Address 1'

Rows Affected:  1

Execute tick count 0

After each user's insertion pFIBDataSet1 executed a query from InsertSQL, and it used the set values as parameters. For example if we try to change a record with Id = 2, SQL Monitor will hook the following query:

pFIBDataSet1: [Execute] UPDATE "Simple Table" SET 

    "Id" = ?"Id",

    "First_Name" = ?"First_Name",

    "Last_Name" = ?"Last_Name",

    "Address" = ?"Address"

 WHERE     

            "Id" = ?"OLD_Id"

    

  Id = 2

  First_Name = 'Name 2 - Changed'

  Last_Name = 'Last Name 2'

  Address = 'Address 2'

  OLD_Id = 2

Rows Affected:  1

Execute tick count 0

As you see, pFIBDataSet1 send to the server all record fields though in reality only the First_Name field was changed. It is easy to imagine, that in a multi-user environment with many fields in a table (especially string ones) such an approach will cause much superfluous net traffic. 

You can eliminate this defect if you use a query generator built in TpFIBDataSet. For this you will have to add some options to AutoUpdateOptions: first set AutoReWriteSQLs and CanChangeSQLs to True, UpdateTableName equal "Simple Table", and UpdateOnlyModifiedFields - True. These options will let pFIBDataSet1 generate modifying queries each time after changing a record. And this query will add only field with really changed values. Run the application and try to change the “First_Name” field of the record with Id = 3. In Memo1 you will see a query executed after such a change:

pFIBDataSet1: [Execute] Update "Simple Table" Set

     "First_Name"=?"NEW_First_Name"

where "Simple Table"."Id"=?"OLD_Id"

 

  NEW_First_Name = 'Name 3 - Changed'

  OLD_Id = 3

Rows Affected:  1

Execute tick count 0

If you use this approach, the economy of net traffic becomes obvious.

Using poRefreshAfterPost option in TpFIBDataSet Options 

The TpFIBDataset component has a special RefreshSQL property intended for refreshing the record, which has just been changed. Imagine a situation when an AFTER UPDATE trigger changing the Last Name field is set to our table from the example above. When a user edits a record and pFIBDataSet1 executes a corresponding UpdateSQL, the trigger also edits the record. After this TpFIBDataset executes a query from RefreshSQL that returns only one current record. For example if we have a look at a query generated with SQL Generator, so RefreshSQL will look like:

SELECT'

    Sim."Id",

    Sim."First_Name",

    Sim."Last_Name",

    Sim."Address"

FROM

    "Simple Table" Sim

WHERE 

    (    

            Sim."Id" = ?"OLD_Id"

    )

It is obvious that after execution of the query we will see the changes made by the trigger. But it is also obvious that the execution of this query after any record changing will cause additional network traffic. If you are sure that a table has no triggers, which change field values or a record in your program will not be edited by multiple users, you can deactivate this query by removing the poRefreshAfterPost key from pFIBDataSet1.Options. In this case RefreshSQL will not be executed without manual calling pFIBDataSet1.Refresh in the program. 

So there is no doubt that deactivation of RefreshSQL during work with tables containing a large number of fields may considerably reduce network traffic and speed up work of your application on low-speed channels.

Client BLOB-filters. «Transparent» packing of BLOB-fields.

Many readers may know about blob filters technology in Firebird. These are user’s functions enabling you to handle (that is code/decode, pack, etc) blob-fields on the server transparently for the client application. This may be useful if you need to archive blob-fields in a database, as for this you do not have to change the client program. But this approach will not help you to decrease the net traffic because in any case the server and the application will exchange unpacked fields. 

FIBPlus has a mechanism of client blob-filters, which is very similar to that in Firebird. An advantage of a local blob-filter is our ability to decrease network traffic of the application considerably if we pack blob-fields before sending them to and then unpack them after getting to the client. This is done by means of registration of two procedures of reading and writing blob-fields in TpFIBDatabase. As a result FIBPlus will automatically use these procedures to handle all blob-fields of the set type in all TpFIBDataSets using one TpFIBDatabase instance. Let's illustrate this mechanism writing an example. 

First we will create a table with blob-fields and a trigger to generate unique values of the primary key:

CREATE TABLE "BlobTable" (

    "Id" INTEGER NOT NULL,

    "BlobText" BLOB sub_type -15 segment size 1);

 

ALTER TABLE "BlobTable" ADD CONSTRAINT "PK_BlobTable" PRIMARY KEY ("Id");

Notice that sub_type must have a negative value! Now place the following components on the form:

pFIBDataSet1: TpFIBDataSet;

pFIBTransaction1: TpFIBTransaction;

pFIBDatabase1: TpFIBDatabase;

DataSource1: TDataSource;

DBGrid1: TDBGrid;

DBMemo1: TDBMemo;

Button1: TButton;

OpenDialog1: TOpenDialog;

Link FIBPlus components and generate queries for pFIBDataSet1 (this time only for the “BlobTable” table) with SQL Generator. We will get the following form: 

We will write a handler of pressing the button:

procedure TForm1.Button1Click(Sender: TObject);

var S: TStream;

    FileS: TFileStream;

begin

  if not OpenDialog1.Execute then exit;

  pFIBDataSet1.Append;

 

  S := pFIBDataSet1.CreateBlobStream(pFIBDataSet1.FieldByName('BlobText'), bmReadWrite);

  FileS := TFileStream.Create(OpenDialog1.FileName, fmOpenRead);

  S.CopyFrom(FileS, FileS.Size);

  FileS.Free;

  S.Free;

  pFIBDataSet1.Post;

end;

Now we will create functions of packing/unpacking blob-fields:

procedure PackBuffer(var Buffer: PChar; var BufSize: LongInt);

var srcStream, dstStream: TStream;

begin

  srcStream := TMemoryStream.Create;

  dstStream := TMemoryStream.Create;

  try

    srcStream.WriteBuffer(Buffer^, BufSize);

    srcStream.Position := 0;

    GZipStream(srcStream, dstStream, 6);

    srcStream.Free;

    srcStream := nil;

    BufSize := dstStream.Size;

    dstStream.Position := 0;

    ReallocMem(Buffer, BufSize);

    dstStream.ReadBuffer(Buffer^, BufSize);

  finally

    if Assigned(srcStream) then srcStream.Free;

    dstStream.Free;

  end;

end;

 

procedure UnpackBuffer(var Buffer: PChar; var BufSize: LongInt);

var srcStream,dstStream: TStream;

begin

  srcStream := TMemoryStream.Create;

  dstStream := TMemoryStream.Create;

  try

    srcStream.WriteBuffer(Buffer^, BufSize);

    srcStream.Position := 0;

    GunZipStream(srcStream, dstStream);

    srcStream.Free;

    srcStream:=nil;

    BufSize := dstStream.Size;

    dstStream.Position := 0;

    ReallocMem(Buffer, BufSize);

    dstStream.ReadBuffer(Buffer^, BufSize);

  finally

    if assigned(srcStream) then srcStream.Free;

    dstStream.Free;

  end;

end;

Do not forget to add two modules to the section uses: zStream, IBBlobFilter. The first is intended for making archives of data and the second is included in FIBPlus and controls blob-filters. Now you only have to register blob-filters. This is done by calling the RegisterBlobFilter function. The value of the first parameter is a type of a blob-field (in our case it is –15), and the second and third parameters are functions of coding and decoding of the blob-field:

procedure TForm1.FormCreate(Sender: TObject);

begin

  pFIBDatabase1.RegisterBlobFilter(-15, @PackBuffer, @UnpackBuffer);

 

  pFIBDatabase1.Connected := True;

  pFIBDataset1.Active := True;

end;

Run our application, delete records it has already contained and add new ones. You will not see any difference but if you look what is really saved in blob-fields, you will see that all the data are archived: 

So, if the application sends to (and gets from) the server already archived blobs, and this can considerably decrease network traffic! Of course you can pack blob-fields without using the above-described mechanism of blob-filters. For example you can compress a field in the Button1Click procedure before saving it and then decompress in the AfterScroll handler or anything like that. But, firstly, use of centralized mechanism greatly simplifies your code (as blob fields are handled imperceptibly for the rest parts of the program) and secondly it helps to avoid commonplace errors (when there are packed blob fields in one part on the program and no ones in another).

FIBPlus Repository: Using a database as storage of queries and client options

Sooner or later most developers start making their applications as simple as possible for some modifications without recompilation of the program. If this concerns work with databases, so as a rule, the developers either begin keeping queries in external text files or at least saving user interface settings in external configuration files. However this approach has a disadvantage as you have to copy settings to all workstations after each their change. So the next stage of creation of more flexible multi-user application is to keep such stuff in the database. Using Firebird and InterBase with FIBPlus we have a ready mechanism called FIBPlus Repository.

Saving and using visual field settings in a database

The simplest ability of FIBPlus Repository is an ability to save such field settings as Display Label or Display Format in the database and then automatically get these settings to the application from the database without writing additional code at all.

We will put on the form the following main components: pFIBDatabase1: TpFIBDatabase, pFIBDataSet: TpFIBDataSet, pFIBTransaction: TpFIBTransaction, DataSource1: TDataSource, DBGrid1: TDBGrid. In the previous articles we have written about how to set parameters of database connection in pFIBDatabase1, to link FIBPlus components with one another, to link DataSource1 and pFIBDataSet, etc so we will not write about this again. The only thing we will mention is that in our examples we will use a standard database Employee.gdb, which is distributed with Borland InterBase.

Set the following query for pFIBDataSet1.SelectSQL:

SELECT

    EMP.EMP_NO,

    EMP.FIRST_NAME,

    EMP.LAST_NAME,

    EMP.PHONE_EXT,

    EMP.HIRE_DATE,

    EMP.DEPT_NO,

    EMP.JOB_CODE,

    EMP.JOB_GRADE,

    EMP.JOB_COUNTRY,

    EMP.SALARY,

    EMP.FULL_NAME

FROM

    EMPLOYEE EMP

And generate modifying queries (UpdateSQL, DeleteSQL и InsertSQL) with SQL Generator supplied with FIBPlus. Now let’s run the application:

As you see, the columns in DBGrid1 have the same name as physical fields in the database. Of course it is not convenient for users, especially if we want to use national symbols in the column names. This happens because in spite of SQL Dialect 3, InterBase has problems with use of national symbols in field names. Delphi (and BCB) enables us to set such field property as DisplayLabel. This property is shown in the DBGrid1 headers. During work in design-time we can create examples of fields, set DisplayLabel values for each of them in Object Inspector and get column names, understandable for users, in DBGrid1. But if during work with the application we will have to change the database structure, we will also have to do all the work again: to delete old fields in pFIBDataSet1 in design-time, add new ones, and again set DisplayLabel, etc. And if you just want to rename a field, you will have to recompile the application and give it to all users in the net again. So as you see every simple operation requires trivial and unnecessary work! You can avoid all these things. FIBPlus enables us to save fields settings in the database. After you change the settings in the database, all new-connected applications will automatically apply them without any changes of code at all. Now let's demonstrate each step.

Set the key urFieldInfo in UseRepository property of pFIBDataBase1. Then press the right button on pFIBDatabase1 and call the FIBPlus Repository field editor:

In the main database FIBPlus Repository creates additional table and it needs our evident permission for this. Press ОК in the appeared dialog to allow FIBPlus creating the additional FIB$FIELDS_INFO table in our database. After the creation we will see a dialog of editing field properties:

As you see, FIBPlus Repository enables us to save such field options as DisplayLabel, Visible, DisplayFormat, EditFormat, DisplayWidth. Pay attention to the Triggered column. Some fields in the tables with the NOT NULL option are intended for automatic filling with database triggers. On getting metadata for generation of modifying queries FIBPlus components set the Required property of all fields with NOT NULL. The program begins to demand obligatory setting of values of such fields from users before sending the query to the server. Such behaviour is reasonable in almost all cases except for the situation when the developer certainly knows that only the trigger has a right to change values of some fields. To be understandable we will show you an example. Let's suppose that there are two triggers in the Employee table, and they form values of the HIRE_DATE field in Employee:

CREATE TABLE EMPLOYEE (

    EMP_NO       EMPNO NOT NULL,

    FIRST_NAME   FIRSTNAME NOT NULL,

    LAST_NAME    VARCHAR(35) NOT NULL,

    PHONE_EXT    VARCHAR(4) CHARACTER SET NONE,

    HIRE_DATE    DATE DEFAULT 'NOW' NOT NULL,

    DEPT_NO      DEPTNO NOT NULL,

    JOB_CODE     JOBCODE NOT NULL,

    JOB_GRADE    JOBGRADE NOT NULL,

    JOB_COUNTRY  COUNTRYNAME NOT NULL,

    SALARY       SALARY NOT NULL,

    FULL_NAME    COMPUTED BY (last_name || ', ' || first_name)

);

 

CREATE TRIGGER EMPLOYEE_NEW_HIRE_DATE FOR EMPLOYEE

ACTIVE AFTER INSERT POSITION 0

AS

begin

  new.hire_date = 'NOW';

end

 

CREATE TRIGGER EMPLOYEE_NEW_HIRE_DATE_UPD FOR EMPLOYEE

ACTIVE AFTER UPDATE POSITION 0

AS

begin

  new.hire_date = 'NOW';

end

It is obvious that there is no sense for the user to change the values of this field. However the HIRE_DATE field is NOT NULL, so our application will require user to set any NOT NULL value. We can avoid this if we set in FIBPlus Repository that the trigger handles this field.

Now we return to our editor of fields of FIBPlus Repository. If we double click the name of the Employee table in the list, its fields will be added to Repository. Now we will only have to set necessary values. We can even hide unnecessary fields, setting Visible equal 0.

Now we should activate the psApplyRepository option in pFIBDataSet1.PrepareOptions and run the application: 

So as you see all our DisplayLabel setting have been applied. Such approach is also nice because it does not matter, in which queries the Employee table takes part, because the centralized field settings will be automatically applied to the whole application. 

Saving additional field options in a database

Sometimes you need some additional settings for table fields. FIBPlus Repository enables you to operate easily with any additional fields in FIB$FIELDS_INFO. Let's suppose that we want to keep DisplayLabel values in three languages. One DISPLAY_LABEL field will not be enough so we add a couple of columns:

ALTER TABLE FIB$FIELDS_INFO ADD GERMANY_LABEL VARCHAR(20);

ALTER TABLE FIB$FIELDS_INFO ADD RUSSIAN_LABEL VARCHAR(20);

And open the field editor in FIBPlus Repository setting alternative names of DisplayLabel in German and Russian:

Then we add the ComboBox1: TComboBox component on the form, set the ComboBox1.Style property equal csDropDownList and add three items to the Items property:

English

German

Russian

We also add the DataSetsContainer1: TDataSetsContainer component. After this we set the Container property of pFIBDataSet1 equal DataSetsContainer1. And connect the pFIBDataInfo module to the uses list in our module:

implementation

uses pFIBDataInfo;

Write OnClick event handler for ComboBox1:

procedure TForm1.ComboBox1Click(Sender: TObject);

begin

  pFIBDataSet1.CloseOpen(false);

end;

We will also write an OnDataSetEvent event handler of DataSetsContainer1:

procedure TForm1.DataSetsContainer1DataSetEvent(DataSet: TDataSet;

  Event: TKindDataSetEvent);

var  I: integer;

     Dl: string;

begin

 if Event = deAfterOpen then

   with DataSet do

     for I := 0 to (FieldCount - 1) do begin

      case ComboBox1.ItemIndex of

         1 : Dl := GetOtherFieldInfo(Fields[I], 'GERMAN_LABEL');

         2 : Dl := GetOtherFieldInfo(Fields[I], 'RUSSIAN_LABEL');

         else

           Dl := GetOtherFieldInfo(Fields[I], 'DISPLAY_LABEL');

      end;

      if Dl <> '' then Fields[I].DisplayLabel := Dl;

     end;

end;

The GetOtherFieldInfo function refers to FIBPlus Repository and returns a value of a necessary parameter for the given field. In our example we look through all the fields of our pFIBDataSet1 one by one and request either the GERMAN_LABEL parameter or RUSSIAN_LABEL depending on the current language chosen in ComboBox1. If you run the application and try to choose the German language in the ComboBox1 list, you will see how the column headers in DBGrid1 will be replaced. Of course this mechanism can be easily applied for many other table settings, which you want to keep in the database. We have used TDataSetsContainer because this component enables us to keep the code of handling additional fields in FIB$FIELDS_INFO in one place of the application. Connecting any other items of TpFIBDataSet to DataSetsContainer1, you will be automatically able to use additional languages in the whole application.

